
New Models Simulate RF Circuits
Its no news to those who simulate that the accuracy of SPICE is
directly related to the accuracy of the models. What may be news
is that simulation of high frequency circuits well into the gigahertz
range is now possible due to the introduction of some new RF
SPICE models.

To illustrate the improved simulation accuracy, the RF models
were used to analyze a 500 MHz oscillator (Figure 1). The
oscillator generates a relatively high power output with a very
distorted output waveform (typical at power levels over the 1-
2mW range). The simulation goals were to study the start-up
characteristics, oscillating frequency and amplitude, and the
resulting harmonic distortion. Two simulations were run. The first
using the standard Gummel-Poon BJT model and simple inductor
chokes and a second using an improved 2N5109 model along
with a new Intusoft RF bead model from the RF Device Library.

Above approximately 100-200 MHz, the built-in SPICE BJT
model, based on the Gummel Poon model, fails to accurately
predict the real device performance. The BJT must be remodeled
as a subcircuit (Table 1) in order to  accurately model the package
and bond wire parasitics which are of greater significance at
higher frequencies. The improved model includes the package
parasitics and matches the s-parameters up to 2 GHz. The RF
library was created by Analog & RF Models, specialists in the
creation of RF models, for Intusoft [1].

Figure 1, A 500 MHz oscillator and RF Bead (shaded area) test circuit.

LB1, LB2, LS1, and LS2 are
parasitic inductances for the
capacitors. LBX, LCX, and
LEX represent the additional
lead inductance beyond the
plane where the manufac-
turer measured the transistor
s-parameters. LEC and LSP
are RF chokes. The varactor
includes parasitic capaci-
tance and inductance inside
the subcircuit.

Although SPICE does contain a model for a nonlinear inductor it
does not have a built-in bead model. Fortunately, one can be
created using the a nonlinear magnetics model and passive
elements. The new model, as shown in Figures 1 & 2, accurately
simulates the impedance vs. frequency response and the change
in impedance vs. temperature. The change in impedance with DC
bias is also modeled due to the addition of the core model instead
of a simple inductor. The bead model gives a very low DC
impedance while providing a large impedance at higher frequen-
cies. The model used here is for a ferrite bead made of a medium
permeability nickel zinc material from Fair-Rite, P#2743009112.
Other types of beads including beads on leads, wound beads and
surface mount beads can be found in the RF library [2].

Modeling An RF Bead

Starting An Oscillator

Figure 2, The Intusoft RF Bead model simulates the proper impedance vs.
frequency characteristics. The graph displays the response for several devices.

Simulation of oscillators present a variety of challenges, not the
least of which is getting the oscillator to oscillate. When ISSPICE

performs an AC or Transient analysis it first performs a DC
analysis in order to establish the starting initial operating point for
the circuit. If a stable operating point is found, which is the goal of
the DC analysis, the oscillator may not oscillate during the
transient (time domain) analysis unless some random distur-
bance is encountered.

There are a number of ways to start an oscillator; each with
varying results and consequences (Figure 3). The method cho-
sen here was to introduce a voltage pulse into the circuit, specifi-
cally, at the emitter of the transistor (VIN 1 0 PWL...). Another
possible method is to insert a current pulse somewhere in the
resonant portion of the circuit, for example at LT (I1 18 17 PULSE
.01 0). If the DC analysis does not converge, a sign that the circuit

Figure 3, Comparison of the various methods of “thumping” an oscillator to get it
started show a current pulse at LT to be the most effective for this circuit.

is unstable and may want to oscillate without any help, the UIC
keyword can be issued in the .TRAN statement.  For example,
.TRAN .1NS 50NS UIC. This will cause the simulation to proceed
directly to the Transient analysis bypassing the DC analysis. The
.IC and “IC=” parameters can then be used to set initial transient
conditions and unbalance the oscillator. One problem with using
UIC is that no DC operating point will be produced inhibiting study
of the circuit bias. The last method is similar to the first and
involves ramping of the power supply (VCC 8 0 PULSE 0 8V 0
5NS). This method may not work well, however, due to the bypass
capacitors. In general, when ramping a source, make sure to give
the ramp a realistic slope in order to avoid “timestep too small”
errors. The first two methods are the most often recommended as
the other methods may not work or may introduce transient start-
up residues [3].

The oscillator circuit was first simulated with two inductor chokes,
LSP and LEC (Figure 1), and a standard .MODEL statement for
the 2N5109 transistor. In SPICE, the standard representation for
a transistor uses the Gummel-Poon model. Various parameters
in the SPICE .MODEL statement are altered in order to cause the
“generic nature” of the Gummel-Poon template to represent a
particular device.

In a second simulation, the chokes were each replaced with the
new RF bead model. The simple transistor model was replaced
with a subcircuit representation. Since the Gummel-Poon model
can not adequately represent BJT behavior above approximately
200MegHz, a composite model must be assembled. The SPICE
subcircuit, containing a BJT model and various parasitic ele-
ments, is utilized for this purpose.

Circuit Modeling



Page 4

When simulating in SPICE it is best to use a subcircuit represen-
tation for a device rather than forcing model parameters to have
unreasonable values. If model parameters are used outside their
physical bounds, the model may work well in one area, but
incorrectly in another. For example, the device may behave
properly during the AC small signal analysis, but poorly during the
nonlinear transient analysis. Some vendors who produce SPICE
model use this approach and the user should beware. Models for
RF transistors that are going to be used for both linear and
nonlinear analysis can not be produced this way. A subcircuit
representation must be used!

Note: Parasitics for the passive elements were maintained for
both simulation cases.

Figure 4, Comparison of the start-up and power supply waveforms with (Top) and
without (Bottom) the new RF bead model.

In order to isolate VCC and not contaminate the power supply with
the 500 MHz oscillating waveform, adequate bypassing is re-
quired. A voltage generator represents a perfect bypass because
it is zero ohms at all frequencies. This is quite different from the
real world.

As shown in Figure 4, use of an inductor causes a droop in the
VCC voltage. The reason for the droop is that the 170nH repre-
sents a large impedance. As the oscillator starts up, the transistor
wants more current. Because of the large inductance it can't draw
adequate current so it starts to discharge the bypass capacitors.
This appears as a drooping in the VCC power (lower graph). The
bead, on the other hand, has a very low DC impedance and a high
AC impedance. By choosing the proper bead, a frequency re-
sponse can be selected that will block all the AC around the
oscillation frequency. With the bead (upper graph), the VCC line
doesn't droop and shows that the size of the ripple stays the same
revealing the imperfections in the bypassing. Also note that the
oscillation starts slower and does not have quite as much power
out with the bead inserted (to be expected) as it does when the
inductors are used.

The oscillator with no BJT parasitics or beads still oscillates
because it was made to be tolerant of package parasitics, but the
results predicted were inaccurate in several important areas. As
shown in Figure 5, the FFT and transient response of the circuit
with beads and new RF BJT subcircuit model reveals that the
frequency of oscillation is lower and the distortion higher.

From the simulations performed, it is clear that modeling the
proper circuit parasitics is of vital importance, especially at RF
frequencies. It is recommended that initial simulations run for
many cycles in order to verify that stable oscillation is actually
taking place. As for performing an FFT, the ISSPICE .TRAN tstart
parameter can be used to delay the start of data taking until steady
state oscillation has been reached.

With the new bead and BJT models in the RF library, ISSPICE is
able to show the peak component stresses when power is
applied, the transient start-up performance, and the variations in
the power supply. In contrast to the linear analysis programs
commonly used by RF designers, ISSPICE simulations can reveal
many important circuit properties such as efficiency, power dissi-
pation, start up characteristics, and harmonic distortion. Charac-
teristics that would be either difficult or impossible to measure.

Results

Conclusions

Figure 5, Comparison of the frequency spectrum and time waveforms using the new
Intusoft BJT and bead models vs. the Gummel-Poon model and inductor chokes. The
new models give a more accurate prediction of distortion and oscillation frequency.

Table 1, ISSPICE Subcircuit for an RF Transistor & Bead
.SUBCKT QN5109 1 2 3
LC 1 4 0.875E-9
LB 2 6 1.590E-9
LE 5 3 2.650E-9
CC 4 3 1.410E-12
CB 4 6 0.561E-12
Q1 4 6 5 QR33
.MODEL QR33 NPN ( BF=44  VAF=160  VAR=16.0  RC=0.69
+RB=1.57 RE=2.75  IKF=0.28E+00  ISE=0.36E-13  TF=0.111E-09
+TR=0.80E-08  ITF=0.82E-01  VTF=0.66E+01  CJC=2.758E-12
+CJE=1.822E-12  XTI=3.0  NE=1.5  ISC=0.12E-13  EG=1.11
+XTB=1.5  BR=1.14  VJC=0.75  VJE=0.75  IS=0.40E-14
+MJC=0.33  MJE=0.33  XTF=4.0  IKR=0.28E+00  KF=0.1E-14
+NC=1.7  FC=0.50  RBM=1.1  IRB=0.40E-01  XCJC=0.5 )
.ENDS

.SUBCKT  BEAD 1  2
R4 1 2 220  TC=-.00333
C2 1 2 .9PF
RX 3 2 1E12
CB 3 2 7.432N
F1 1 2 VM1 1
G2 2 3 1 2 1
E1 4 2 3 2 1
VM1 4 5
RB 5 2 341.5
RS 5 6 3.7904
VP 7 2 300
D1 6 7 DCLAMP
VN 2 8 300
D2 8 6 DCLAMP

.MODEL DCLAMP D
+CJO=8.2286P VJ=25
.ENDS

4

6

5 3

2

1

Q1 
LB 

CB 

LC 

LE 

CC 



Page 7

References

[1] Analog & RF Models, Bill Sands (602) 575-5323, FAX (602)
297-5160

[2] RF Device Library, Intusoft, 1990

[3] “SIMULATING WITH SPICE”, L.G. Meares, C.E. Hymowtiz, Intusoft,
1988


